



ORF 307

Lecture 3

Chapter 13, Section 1

Portfolio Optimization

Robert Vanderbei

February 9, 2012

Portfolio Optimization: Markowitz Shares the 1990 Nobel Prize



Press Release - The Sveriges Riksbank (Bank of Sweden) Prize in Economic Sciences
in Memory of Alfred Nobel

KUNGL. VETENSKAPSAKADEMIEN THE ROYAL SWEDISH ACADEMY OF SCIENCES

16 October 1990

THIS YEAR'S LAUREATES ARE PIONEERS IN THE THEORY OF FINANCIAL ECONOMICS
AND CORPORATE FINANCE

The Royal Swedish Academy of Sciences has decided to award the 1990 Alfred Nobel Memorial Prize
in Economic Sciences with one third each, to

Professor **Harry Markowitz**, City University of New York, USA,
Professor **Merton Miller**, University of Chicago, USA,
Professor **William Sharpe**, Stanford University, USA,

for their pioneering work in the theory of financial economics.


Harry Markowitz is awarded the Prize for having developed the theory of portfolio choice;
William Sharpe, for his contributions to the theory of price formation for financial assets, the so-called,
Capital Asset Pricing Model (CAPM); and
Merton Miller, for his fundamental contributions to the theory of corporate finance.

Summary

Financial markets serve a key purpose in a modern market economy by allocating productive resources among various areas of production. It is to a large extent through financial markets that saving in different sectors of the economy is transferred to firms for investments in buildings and machines. Financial markets also reflect firms' expected prospects and risks, which implies that risks can be spread and that savers and investors can acquire valuable information for their investment decisions.


The first pioneering contribution in the field of financial economics was made in the 1950s by Harry Markowitz who developed a theory for households' and firms' allocation of financial assets under uncertainty, the so-called theory of portfolio choice. This theory analyzes how wealth can be optimally invested in assets which differ in regard to their expected return and risk, and thereby also how risks can be reduced.

Historical Data—Some ETF Prices



Notation: $S_j(t)$ = share price for investment j at time t .

Return Data: $R_j(t) = S_j(t)/S_j(t - 1)$



Important observation: *volatility* is easy to see, *mean return* is lost in the noise.

Risk vs. Reward

Reward—estimated using historical means:

$$\text{reward}_j = \frac{1}{T} \sum_{t=1}^T R_j(t).$$

Risk—Markowitz defined risk as the variability of the returns as measured by the historical variances:

$$\text{risk}_j = \frac{1}{T} \sum_{t=1}^T (R_j(t) - \text{reward}_j)^2.$$

However, to get a linear programming problem (and for other reasons) we use the sum of the absolute values instead of the sum of the squares:

$$\text{risk}_j = \frac{1}{T} \sum_{t=1}^T |R_j(t) - \text{reward}_j|.$$

Why Make a Portfolio? ... Hedging


Investment A: up 20%, down 10%, equally likely—a risky asset.

Investment B: up 20%, down 10%, equally likely—another risky asset.

Correlation: up years for A are down years for B and vice versa.

Portfolio—half in A, half in B: up 5% every year! No risk!

Return Data: 50 days around 01/01/2005



Note: Not much *negative* correlation is price fluctuations. An up day is an up day and a down day is a down day.

Portfolios

Fractions: x_j = fraction of portfolio to invest in j .

Portfolio's Historical Returns:

$$R(t) = \sum_j x_j R_j(t)$$

Portfolio's Reward:

$$\text{reward}(x) = \frac{1}{T} \sum_{t=1}^T R(t) = \frac{1}{T} \sum_{t=1}^T \sum_j x_j R_j(t)$$

Portfolio's Risk:

$$\begin{aligned}\text{risk}(x) &= \frac{1}{T} \sum_{t=1}^T |R(t) - \text{reward}(x)| \\ &= \frac{1}{T} \sum_{t=1}^T \left| \sum_j x_j R_j(t) - \frac{1}{T} \sum_{s=1}^T \sum_j x_j R_j(s) \right| \\ &= \frac{1}{T} \sum_{t=1}^T \left| \sum_j x_j \left(R_j(t) - \frac{1}{T} \sum_{s=1}^T R_j(s) \right) \right| \\ &= \frac{1}{T} \sum_{t=1}^T \left| \sum_j x_j (R_j(t) - \text{reward}_j) \right|\end{aligned}$$

A Markowitz-Type Model

Decision Variables: the fractions x_j .

Objective: maximize return, minimize risk.

Fundamental Lesson: can't simultaneously optimize two objectives.

Compromise: set an upper bound μ for risk and maximize reward subject to this bound constraint:

- Parameter μ is called **risk aversion parameter**.
- Large value for μ puts emphasis on reward maximization.
- Small value for μ puts emphasis on risk minimization.

Constraints:

$$\frac{1}{T} \sum_{t=1}^T \left| \sum_j x_j (R_j(t) - \text{reward}_j) \right| \leq \mu$$
$$\sum_j x_j = 1$$
$$x_j \geq 0 \quad \text{for all } j$$

Optimization Problem

$$\begin{aligned} \text{maximize} \quad & \frac{1}{T} \sum_{t=1}^T \sum_j x_j R_j(t) \\ \text{subject to} \quad & \frac{1}{T} \sum_{t=1}^T \left| \sum_j x_j (R_j(t) - \text{reward}_j) \right| \leq \mu \\ & \sum_j x_j = 1 \\ & x_j \geq 0 \quad \text{for all } j \end{aligned}$$

Because of absolute values not a linear programming problem.

Easy to convert...

A Linear Programming Formulation

$$\begin{aligned} &\text{maximize} && \frac{1}{T} \sum_{t=1}^T \sum_j x_j R_j(t) \\ &\text{subject to} && -y_t \leq \sum_j x_j (R_j(t) - \text{reward}_j) \leq y_t && \text{for all } t \\ &&& \frac{1}{T} \sum_{t=1}^T y_t \leq \mu \\ &&& \sum_j x_j = 1 \\ &&& x_j \geq 0 && \text{for all } j \end{aligned}$$

AMPL: Model

```
set A; # asset categories
set T; # dates

param n := card(T);
param risk;
param R {T,A};
param mean {j in A} := ( sum{i in T} R[i,j] )/n;
param Rtilde {i in T, j in A} := R[i,j] - mean[j];

var x{A} >=0;
var dev{T} >= 0;

maximize reward: sum{j in A} mean[j]*x[j] ;

s.t. risk_bound: sum{i in T} dev[i] / n <= risk;
s.t. tot_mass: sum{j in A} x[j] = 1;
s.t. dev_lo_bnd {i in T}: -dev[i] <= sum{j in A} Rtilde[i,j]*x[j];
s.t. dev_up_bnd {i in T}: sum{j in A} Rtilde[i,j]*x[j] <= dev[i];
```

AMPL: Data, Solve, and Print

```
data;

set A := xlu xlb xli xlv xlf xle mdy xlk xly xlp qqqq spy;
set T := include 'dates.out';

param R: xlu xlb xli xlv xlf xle mdy xlk xly xlp qqqq spy:=
include 'amplreturn3.data' ;

printf {j in A}: "%10.7f %10.5f \n", mean[j], sum{i in T} abs(Rtilde[i,j])/n > "assets";

for {k in 0..200 by 15} {
    display k;

    let risk := (k/200)*0.007 + (1-k/200)*0.017;
    solve;

    printf: "%7.4f \n", risk > "portfolio";
    printf {j in A: x[j] > 0.001}: "%45s %6.3f \n", j, x[j] > "portfolio";
    printf: "
                %6.3f %6.3f \n",
        sum{j in A} mean[j]*x[j],
        sum{i in T} abs(sum{j in A} Rtilde[i,j]*x[j]) / n > "portfolio";

    printf: "%10.7f %10.5f \n",
        sum{j in A} mean[j]*x[j],
        sum{i in T} abs(sum{j in A} Rtilde[i,j]*x[j]) / n > "eff_front";
}
```

Efficient Frontier


Varying risk bound μ produces the so-called *efficient frontier*.

Portfolios on the efficient frontier are reasonable.

Portfolios not on the efficient frontier can be strictly improved.

XLU	XLB	XLV	XLF	XLE	MDY	XLK	XLY	XLP	QQQ	SPY	Reward	Risk
				1.000							1.110	0.0137
				1.000							1.110	0.0137
				1.000							1.110	0.0137
				1.000							1.110	0.0137
				1.000							1.110	0.0137
0.050				0.950							1.107	0.0132
0.138				0.862							1.101	0.0125
0.233				0.767							1.096	0.0117
0.329				0.666				0.006			1.090	0.0110
0.332				0.589				0.078			1.083	0.0102
0.329				0.509				0.162			1.076	0.0095
0.311				0.424	0.005			0.259			1.069	0.0087
0.308				0.319	0.015			0.359			1.060	0.0080
0.273		0.067		0.205				0.456			1.048	0.0072

Efficient Frontier



Getting Started With AMPL on a PC

Download PUTTY from

<http://www.chiark.greenend.org.uk/~sgtatham/putty/download.html>

Under “Start”, “Programs”, click on “Putty” under, you guessed it, “Putty”.

For “hostname” enter “hats.princeton.edu”.

Log in with your usual user id and password.

For a text editor, I recommend PICO, so, just type “pico”.